
	Reason for change:
	In the current TTCN implementation for 7.1.4.1, In step 5 , RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND is sent by SS.

In step 5 , cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig) is used, which is having Complete nonCriticalExtension IE set as Omit. [ each element inside  nonCriticalExtension is left uninitialized ]

While filling dedicatedNAS_MessageList in the altstep a_AspFromTestcaseHandler_NRNG , only  dedicatedNAS_MessageList  is getting assigned to NAS_DedicatedMsg  nonCriticalExtension.dedicatedNAS_MessageList [I] :=                   v_NAS_DedicatedMsg[I];

Because of the above assignment, only nonCriticalExtension.dedicatedNAS_MessageList is set with values where as other below IEs are left uninitilized .
v_V1530Ext.fullConfig ,
v_V1530Ext.masterCellGroup ,
v_V1530Ext.masterKeyUpdate , 
v_V1530Ext.dedicatedSIB1_Delivery ,     v_V1530Ext.dedicatedSystemInformationDelivery , 
v_V1530Ext.nonCriticalExtension ,
v_V1530Ext.otherConfig  

This implementation is creating runtime issue only when reconfiguration is sent with NAS Message Piggybacking. Otherwise nonCriticalExtension altogether is set to Omit ; In case of sending Piggybacked Nas Request , this issue is exposed as dedicatedNAS_MessageList inside nonCriticalExtension is set , and other fields are left uninitialized

0. [bookmark: _Toc69124791]Correction to function fl_TC_7_1_4_1_NR_TestBody
	Function name
	function fl_TC_7_1_4_1_NR_TestBody

	Reason for change
	In the current TTCN implementation for 7.1.4.1, In step 5 , RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND is sent by SS.

In step 5 , cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig) is used, which is having Complete nonCriticalExtension IE set as Omit. [ each element of  nonCriticalExtension is left uninitialized ]

While filling dedicatedNAS_MessageList in the altstep a_AspFromTestcaseHandler_NRNG , only  dedicatedNAS_MessageList  is getting assigned to NAS_DedicatedMsg  nonCriticalExtension.dedicatedNAS_MessageList [I] :=                   v_NAS_DedicatedMsg[I];

Because of the above assignment, only nonCriticalExtension.dedicatedNAS_MessageList is set with values where as other below IEs are left uninitilized .
v_V1530Ext.fullConfig ,
v_V1530Ext.masterCellGroup ,
v_V1530Ext.masterKeyUpdate , 
v_V1530Ext.dedicatedSIB1_Delivery ,     v_V1530Ext.dedicatedSystemInformationDelivery , 
v_V1530Ext.nonCriticalExtension ,
v_V1530Ext.otherConfig  
This implementation is creating issue only when reconfiguration is sent with NAS Message Piggybacking. Otherwise nonCriticalExtension altogether is set to Omit , in case of sending Piggybacked Nas Request , this issue is exposed as dedicatedNAS_MessageList inside nonCriticalExtension is set , and other fields are left uninitialized

	Summary of change
	Uninitilized IEs in nonCriticalExtension are set to Omit

	TTCN module
	ttcn\develop\NR5GC\7_1_4\SDAP_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function fl_TC_7_1_4_1_NR_TestBody(integer p_FirstDRB,
                                     integer p_SecondDRB,
                                     GSM_MobilityInfo_Type p_GSM_MobilityInfo_Type) runs on NR_BASE_PTC
  {
    var template (present) SDAP_PDU_Type v_ExpectedPdu;
    var octetstring v_IPDataQFI1 := f_SdapTests_CreateDatagram("2"); // Packet with remote address matching filter #2
    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
    var octetstring v_IPDataQFI4 := f_SdapTests_CreateDatagram("3a"); // Packet with remote address matching filter #3a
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var integer v_PduSessionId := oct2int(p_GSM_MobilityInfo_Type.SessionId);  // @sic R5s191119, R5s200564 sic@
    var boolean v_PDU := false; // @sic R5s191119 sic@; boolean to indicate if the SDAP PDU has been received
    var boolean v_EndMarker := false; // @sic R5s191119 sic@; boolean to indicate if the End-Marker has been received
    var boolean v_RrcMsg := false; // @sic R5s191119 sic@; boolean to indicate if the RRC message has been received
    var boolean v_NasMsg := false; // @sic R5s191119 sic@; boolean to indicate if the NAS message has been received
    var NR_SRB_COMMON_IND v_ReceivedAsp; // @sic R5s191119 sic@
    
    // @siclog "Step 1" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=1.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI1, v_IPDataQFI1);
        
    // @siclog "Step 2" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping flow with QFI=1?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI1),v_IPDataQFI1);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=1, RQI=0, QFI=5.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI1, tsc_RQI0, tsc_QFI5, v_IPDataQFI5);
    
    // EXCEPTION: In parallel to the event described in step 4 the events specified in Table 7.1.4.1.3.2-2 shall take place.
    // @siclog "Table 7.1.4.1.3.2-2, Step 1" siclog@
    // @siclog "Step 4" siclog@

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5),v_IPDataQFI5);
    // @sic R5s191119 sic@
    do {
      alt {
        [not v_PDU] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}))
          { // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=5?
            v_PDU := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI5)}))
          { // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=5?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
      }
    }
    while (not(v_PDU and v_EndMarker));
    
    // @siclog "Step 5" siclog@
    // The SS transmits an RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND
    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {4}), // @sic R5s191119 sic@
                                                   p_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300))};
   
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,
                                                         v_DRB_ToAddModList,
                                                         omit,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
                                                         
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand (p_GSM_MobilityInfo_Type.SessionId,
                                                                      -,-,-,-,-,
                                                                      cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("4a")}, '7A'O), // @sic R5s191119 sic@
                                                                      cs_QoSFlowDescr ({cs_QoS_Flow2a}));
   
    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,
                                      tsc_NR_RbId_SRB1,
                                      cs_TimingInfo_Now,
                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig)),
                                      {cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)}));
    
    // EXCEPTION: In parallel to the event described in step 6 the events specified in Tables 7.1.4.1.3.2-3 and 7.1.4.1.3.2-4 shall take place.
    // @siclog "Table 7.1.4.1.3.3-3, Step 1" siclog@
    // @siclog "Table 7.1.4.1.3.3-4, Step 1" siclog@
    // @siclog "Step 6" siclog@
    // @sic R5s191119 sic@
    v_EndMarker := false; // Reset variable for reusing
    do {
      alt {
        [not v_RrcMsg] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))
          { // The UE transmits an RRCReconfigurationComplete message.
            v_RrcMsg := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI4)}))
          { // Check: Does the UE transmit End-Marker Control PDU on DRB 1 for QFI=4?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
        [not v_NasMsg] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
          { // The UE Transmits PDU SESSION MODIFICATION COMPLETE
            if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {
              v_NasMsg := true;
            }
            else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS message received");
            }
          }
      }
    }
    while (not(v_RrcMsg and v_EndMarker and v_NasMsg));
    
    // @siclog "Step 7" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=4.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI4, v_IPDataQFI4);
    
    // @siclog "Step 8" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=4?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI4),v_IPDataQFI4);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
      
  }



After Change:
	function fl_TC_7_1_4_1_NR_TestBody(integer p_FirstDRB,
                                     integer p_SecondDRB,
                                     GSM_MobilityInfo_Type p_GSM_MobilityInfo_Type) runs on NR_BASE_PTC
  {
    var template (present) SDAP_PDU_Type v_ExpectedPdu;
    var octetstring v_IPDataQFI1 := f_SdapTests_CreateDatagram("2"); // Packet with remote address matching filter #2
    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
    var octetstring v_IPDataQFI4 := f_SdapTests_CreateDatagram("3a"); // Packet with remote address matching filter #3a
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var integer v_PduSessionId := oct2int(p_GSM_MobilityInfo_Type.SessionId);  // @sic R5s191119, R5s200564 sic@
    var boolean v_PDU := false; // @sic R5s191119 sic@; boolean to indicate if the SDAP PDU has been received
    var boolean v_EndMarker := false; // @sic R5s191119 sic@; boolean to indicate if the End-Marker has been received
    var boolean v_RrcMsg := false; // @sic R5s191119 sic@; boolean to indicate if the RRC message has been received
    var boolean v_NasMsg := false; // @sic R5s191119 sic@; boolean to indicate if the NAS message has been received
    var NR_SRB_COMMON_IND v_ReceivedAsp; // @sic R5s191119 sic@
    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;  
   
    v_V1530Ext.fullConfig := omit;
    v_V1530Ext.masterCellGroup := omit;
    v_V1530Ext.masterKeyUpdate := omit;
    v_V1530Ext.dedicatedSIB1_Delivery := omit;
    v_V1530Ext.dedicatedSystemInformationDelivery := omit;
    v_V1530Ext.nonCriticalExtension := omit;
    v_V1530Ext.otherConfig := omit; 
    v_V1530Ext.dedicatedNAS_MessageList := omit;
      
    
    // @siclog "Step 1" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=1.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI1, v_IPDataQFI1);
        
    // @siclog "Step 2" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping flow with QFI=1?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI1),v_IPDataQFI1);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=1, RQI=0, QFI=5.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI1, tsc_RQI0, tsc_QFI5, v_IPDataQFI5);
    
    // EXCEPTION: In parallel to the event described in step 4 the events specified in Table 7.1.4.1.3.2-2 shall take place.
    // @siclog "Table 7.1.4.1.3.2-2, Step 1" siclog@
    // @siclog "Step 4" siclog@

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5),v_IPDataQFI5);
    // @sic R5s191119 sic@
    do {
      alt {
        [not v_PDU] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}))
          { // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=5?
            v_PDU := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI5)}))
          { // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=5?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
      }
    }
    while (not(v_PDU and v_EndMarker));
    
    // @siclog "Step 5" siclog@
    // The SS transmits an RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND
    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {4}), // @sic R5s191119 sic@
                                                   p_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300))};
   
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,
                                                         v_DRB_ToAddModList,
                                                         omit,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
                                                         
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand (p_GSM_MobilityInfo_Type.SessionId,
                                                                      -,-,-,-,-,
                                                                      cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("4a")}, '7A'O), // @sic R5s191119 sic@
                                                                      cs_QoSFlowDescr ({cs_QoS_Flow2a}));
   
    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,
                                      tsc_NR_RbId_SRB1,
                                      cs_TimingInfo_Now,
                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, - ,v_V1530Ext)),
                                      {cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)}));
    
    // EXCEPTION: In parallel to the event described in step 6 the events specified in Tables 7.1.4.1.3.2-3 and 7.1.4.1.3.2-4 shall take place.
    // @siclog "Table 7.1.4.1.3.3-3, Step 1" siclog@
    // @siclog "Table 7.1.4.1.3.3-4, Step 1" siclog@
    // @siclog "Step 6" siclog@
    // @sic R5s191119 sic@
    v_EndMarker := false; // Reset variable for reusing
    do {
      alt {
        [not v_RrcMsg] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))
          { // The UE transmits an RRCReconfigurationComplete message.
            v_RrcMsg := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI4)}))
          { // Check: Does the UE transmit End-Marker Control PDU on DRB 1 for QFI=4?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
        [not v_NasMsg] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
          { // The UE Transmits PDU SESSION MODIFICATION COMPLETE
            if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {
              v_NasMsg := true;
            }
            else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS message received");
            }
          }
      }
    }
    while (not(v_RrcMsg and v_EndMarker and v_NasMsg));
    
    // @siclog "Step 7" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=4.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI4, v_IPDataQFI4);
    
    // @siclog "Step 8" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=4?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI4),v_IPDataQFI4);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
      
  }



